apply plugin: 'com.gorylenko.gradle-git-properties'

dependencies {
    api "io.projectreactor.netty:reactor-netty-http"
    api "io.projectreactor.addons:reactor-extra"
    api "com.fasterxml.jackson.core:jackson-databind:$jackson_databind_version"
    api "com.fasterxml.jackson.datatype:jackson-datatype-jdk8:$jackson_datatype_jdk8_version"
    api "com.github.ben-manes.caffeine:caffeine:$caffeine_version"

    api "com.discord4j:stores-api:$storesVersion"

    testImplementation "org.junit.jupiter:junit-jupiter-engine:$junit_version"
    testImplementation "io.projectreactor:reactor-test"
    testImplementation "ch.qos.logback:logback-classic:$logback_version"
}

gitProperties {
    gitPropertiesDir = file("${project.getLayout().getBuildDirectory().get()}/resources/main/discord4j/common")
    customProperty 'git.commit.id.describe', { it.describe(tags: true) }
    customProperty 'application.name', project_name
    customProperty 'application.description', project_description
    customProperty 'application.url', project_url
}

jar {
    manifest {
        attributes 'Implementation-Title': base.archivesName.get(),
                'Implementation-Version': archiveVersion,
                'Automatic-Module-Name': 'discord4j.common'
    }
}
